Isothermal coordinates and conformal maps

We compute the triple compositions (St)o(U)o(X) for isothermal parameterizations of some minimal surfaces, and try to identify whether the map is indeed analytic. (It should be!)

```
> restart:
  with(plots):
  with(linalg):
Warning, new definition for norm
Warning, new definition for trace
```

1) Some procedures to help compute the Gauss map:

```
> #dot product
  dp:=proc(X,Y)
  X[1]*Y[1]+X[2]*Y[2]+X[3]*Y[3];
  end:
> #2-norm
  nrm:=proc(X)
  sqrt(dp(X,X));
  end:
> #cross product:
  xp:=proc(X,Y)
  local a,b,c;
  a:=X[2]*Y[3]-X[3]*Y[2];
  b:=X[3]*Y[1]-X[1]*Y[3];
  c:=X[1]*Y[2]-X[2]*Y[1];
  [a,b,c];
  end:
> #Derivative matrix for mapping X:
  DXq:=proc(X)
  local Xu,Xv;
  Xu:=matrix(3,1,[diff(X[1],u),diff(X[2],u),diff(X[3],u)]);
  Xv:=matrix(3,1,[diff(X[1],v),diff(X[2],v),diff(X[3],v)]);
  simplify(augment(Xu,Xv));
  end:
> #unit normal:
  N:=proc(X)
  local Y,Z,s;
  Y:=DXq(X);
  Z:=xp(col(Y,1),col(Y,2));
  s:=nrm(Z);
  simplify(evalm((1/s)*Z));
  end:
```

2) Add a procedure for stereographic projection:

```
> St:=proc(X)
  simplify([X[1]/(1-X[3]),X[2]/(1-X[3]),0]);
  end:
```

3) Check the triple composition for the Catenoid and Enneper's Surface !

```
> assume(u,real):
```

```
  assume(v,real):
> Cat:=(u,v)->[cos(v)*cosh(u),sin(v)*cosh(u),u]:
     #catenoid parameterization
  St(N(Cat(u,v)));
     #can you recognize this?
> Enne:=(u,v)->[u-u^3/3+u*v^2,-v+v^3/3-v*u^2,u^2-v^2]:
> St(N(Enne(u,v)));
     #how about this?
```